Partial Fraction Decomposition

László Szili and János Tóth

The present notebook is an electronic supplement to the authors' paper entitled
Partial fraction decomposition of some meromorphic functions
which appeared in *Annales Univ. Sci. Budapest., Sect. Comp.,* **38** (2012), 93-108.

Partial fraction decomposition occurs twice in a usual curriculum: first, as a technique helping calculate primitive functions (or anti-derivatives) of rational functions, second, as a method to find the inverse Laplace transform of functions.
Solving differential equations is quite common using the notion of Laplace transform, especially in engineering schools. The main advantage of this technique is not in solving simple linear, constant coefficient equations, its merits lies in solving equations containing convolutions or discontinuous terms. Another advantage is (even in the case of the simplest equations) that using the Laplace transform one only determines the unique solution (s)he is interested in, whereas other methods calculate the general solution and they only later specialize it using the given initial and boundary conditions.

Our aim is to show possibilities offered by *Mathematica* for computation of partial fraction decomposition of rational and meromorphic functions.

1. Partial fraction decomposition of rational functions

1.1 Preliminaries

We shall consider polynomials on the complex domain with complex coefficients, i.e. polynomials of the form

$$p(z) := a_0 + a_1 z^1 + \cdots + a_{N-1} z^{N-1} + a_N z^N \quad (z \in \mathbb{C})$$

where N is a fixed positive integer and a_n $(n = 1, 2, \cdots, N)$ are comlex numbers; $a_N \neq 0$.

Let p and q be polynomials without common roots. The function $f := \frac{p}{q}$ is called a *rational function*.

The roots of the denominator polynomial q play an important role in the following. From the Fundamental Theorem of Algebra it follows that q has as many complex roots as its degree is, considering multiplicities.
We will assume that the roots of the polynomial q as well as their multiplicites are known (more precisely *Mathematica* can compute those). The following statement is true:

Theorem.
Let p and q be polynomials without common roots, deg p < deg q and let a_k the roots of q, *a_k having the multiplicity m_k* (k = 1,2, ..., n). *Then there are uniquely determined (complex) numbers $c_{j,k}$* (j = 1,2, ..., m_k; k = 1,2, ..., n) *such that*

$$f(z) = \frac{p(z)}{q(z)} = \sum_{k=1}^{n} \sum_{j=1}^{m_k} c_{j,k} \left(\frac{1}{z - a_k}\right)^j \qquad (z \in \mathbb{C} \setminus \{a_1, \ldots, a_n\}).$$

This is called the ***partial fraction decomposition*** *of the rational function f.*

1.2 A *Mathematica* function

Using the *Mathematica* functions `Solve[]`, `Factor[]` and `Apart[]` we define the following new function:

```
PartFracRat[funval_, var_: z] :=
 Module[{numer = Numerator[funval], denom = Denominator[funval]},
                                   Apart[numer / Factor[denom,
     Extension → ((Re[#] + ⅈ Im[#] &) /@ (var /. Solve[denom == 0, var]))]]]
```

`Solve[]` attempts to find the complex roots of the denominator. `Factor[]` with the option `Extension` factors the denominator allowing coefficients that are rational combinations of the given algebraic numbers. Finally `Apart[]` gives the partial fraction decomposition of the rational function *p/q.*

$$\text{PartFracRat}\left[\frac{1}{(z+1)^2\,(z-2)^3}\right]$$

$$\frac{1}{9\,(-2+z)^3}-\frac{2}{27\,(-2+z)^2}+\frac{1}{27\,(-2+z)}-\frac{1}{27\,(1+z)^2}-\frac{1}{27\,(1+z)}$$

$$\text{PartFracRat}\left[\frac{1}{z^4+1}\right]$$

$$\frac{1}{2\sqrt{2}\left(\sqrt{2}-(1+i)\,z\right)}+\frac{1}{2\sqrt{2}\left(\sqrt{2}-(1-i)\,z\right)}+\frac{1}{2\sqrt{2}\left(\sqrt{2}+(1-i)\,z\right)}+\frac{1}{2\sqrt{2}\left(\sqrt{2}+(1+i)\,z\right)}$$

2. Partial fraction decomposition of the function $\frac{1}{\sin^r}$

2.1 The MacLaurin expansion of the function F_r

In the partial fraction decomposition of the function $\frac{1}{\sin^r}$ the MacLaurin expansion of the function F_r plays an important rule.

$$F_{r\_}(z\_) := \left(\frac{z}{\sin(z)}\right)^r$$

It is analytic on the circle $|z| < \pi$ and its MacLaurin series is given by

$$F_r(z) = \sum_{j=0}^{+\infty} \frac{F_r^{(j)}(0)}{j!} z^j \qquad (|z| < \pi).$$

The computation of the *j*th derivative of F_r is not too easy. *Mathematica* can help us for specified values of *j* in the following way:

```
Limit[D[F_r[z], {z, 6}], z → 0]
```

$$\frac{1}{63} r \left(16 + 42 r + 35 r^2\right)$$

However, general symbolic result is not provided by the program:

```
Limit[D[F_r[z], {z, j}], z → 0]
```

$$\text{Limit}\left[\partial_{\{z,j\}} (z \,\text{Csc}[z])^r, z \to 0\right]$$

Another way to obtain the MacLaurin expansion of F_r is using the **Series[ ]** *Mathematica* function:

```
MacPol[r_, n_] := Series[F_r[z], {z, 0, n}]
```

For example, if $r = 3$ and $n = 6$ then we obtain that

```
MacPol[3, 6]
```

$$1 + \frac{z^2}{2} + \frac{17 z^4}{120} + \frac{457 z^6}{15120} + O[z]^7$$

Mathematica can do the calculations with a symbolic value of r and with specified values n.
For example, if r is arbitrary and $n = 6$ then we have:

```
MacPol[r, 6]
```

$$1 + \frac{r z^2}{6} + \left(\frac{r}{180} + \frac{r^2}{72}\right) z^4 + \frac{\left(16 r + 42 r^2 + 35 r^3\right) z^6}{45360} + O[z]^7$$

The coefficients of the MacLaurin expansion can be obtained in the following way:

```
SeriesCoefficient[F_3[z], {z, 0, 4}]
```

$$\frac{17}{120}$$

```
SeriesCoefficient[F_r[z], {z, 0, 4}]
```

$$\frac{r}{180} + \frac{r^2}{72}$$

Mathematica also gives the coefficients, if n is an arbitrary (natural) number, but r is a specified (not too large) natural number.
For example, if r=1 then we obtain that:

```
Simplify[Assuming[n > 0 && n ∈ Integers, SeriesCoefficient[F1[z], {z, 0, n}]]]
```

$$\frac{(2\,i)^n\,\text{BernoulliB}\left[n,\ \frac{1}{2}\right]}{n!}$$

where `BernoulliB [n, x]` is the Bernoulli polynomial.

This works for $r = 2$, too :

```
Simplify[Assuming[n > 0 && n ∈ Integers, SeriesCoefficient[F2[z], {z, 0, n}]]]
```

$$-\frac{(2\,i)^n\,(-1+n)\,\text{BernoulliB}[n]}{n!}$$

where `BernoulliB[n]` is the *n*th Bernoulli number.

Here are the first few Bernoulli numbers.

```
TableForm[BernoulliB /@ Range[0, 8], TableHeadings → {Range[0, 8], None}]
```

0	1
1	$-\frac{1}{2}$
2	$\frac{1}{6}$
3	0
4	$-\frac{1}{30}$
5	0
6	$\frac{1}{42}$
7	0
8	$-\frac{1}{30}$

However, general symbolic result is not provided by the program.

```
SeriesCoefficient[Fr[z], {z, 0, n}]
```

```
SeriesCoefficient[(z Csc[z])^r, {z, 0, n}]
```

In the next Section we shall give computable recursive relations for the coefficients of the MacLaurin series of the function F_r to extend the capabilities of *Mathematica.*

2.2 Recursive formulas for the coefficients of the MacLaurin series of F_r

The function F_r is an *even* analytic function on the circle $|z| < \pi$, therefore we have

$$F_r^{(2j+1)}(0) = 0 \quad (j \in \mathbb{N}_0 := \{0, 1, 2, \ldots\}).$$

Theorem 1 states that for the coefficients $F(r, 2j) := F_r^{(2j)}(0)$ we have the following recursive relations :

```
F[r_, 0] = 1;
F[r_, j_] :=
 Simplify[r/j Sum[(-1)^(j/2+l-1) Binomial[j, 2 l] 2^(j-2 l) F[r, 2 l] BernoulliB[j - 2 l], {l, 0, j/2-1}]]
```

$$F[r,j] := \mathrm{Simplify}\left[\frac{r}{j}\sum_{l=0}^{j/2-1}(-1)^{j/2+l-1}\,\mathrm{Binomial}[j,2l]\,2^{j-2l}\,F[r,2l]\,\mathrm{BernoulliB}[j-2l]\right]$$

where **Binomial[j,l]** gives the binomial coefficient $\binom{j}{l}$.

The first few coefficients from the recurrence formula (3) are as follows:

```
TableForm[(F[r, #1] &) /@Range[0, 12], TableHeadings → {Range[0, 12], None}]
```

0	1
1	0
2	$\frac{r}{3}$
3	0
4	$\frac{1}{15}\, r\,(2+5\,r)$
5	0
6	$\frac{1}{63}\, r\,(16+42\,r+35\,r^2)$
7	0
8	$\frac{1}{135}\, r\,(144+404\,r+420\,r^2+175\,r^3)$
9	0
10	$\frac{1}{99}\, r\,(768+2288\,r+2684\,r^2+1540\,r^3+385\,r^4)$
11	0
12	$\frac{r\,(1\,061\,376+3\,327\,584\,r+4\,252\,248\,r^2+2\,862\,860\,r^3+1\,051\,050\,r^4+175\,175\,r^5)}{12\,285}$

2.3 The principal part of $\frac{1}{\sin^r}$ at the rth order pole 0

The principal part $G(r, w) := G_{0,r}(w)$ is given by formula (4):

```
G[r_, w_] := Sum[F[r, 2 j]/(2 j)! w^(r-2 j), {j, 0, Floor[(r - 1) / 2]}]
```

$$G[r,w] := \mathrm{Sum}\left[\frac{F[r,2j]}{(2j)!}\,w^{r-2j},\ \{j,0,\mathrm{Floor}[(r-1)/2]\}\right]$$

The results for the first few values of r are as follows:

```
TableForm[G[#, 1 / z] & /@ Range[0, 12], TableHeadings → {Range[0, 12], None}]
```

0	0
1	$\frac{1}{z}$
2	$\frac{1}{z^2}$
3	$\frac{1}{z^3}+\frac{1}{2 z}$
4	$\frac{1}{z^4}+\frac{2}{3 z^2}$
5	$\frac{1}{z^5}+\frac{5}{6 z^3}+\frac{3}{8 z}$
6	$\frac{1}{z^6}+\frac{1}{z^4}+\frac{8}{15 z^2}$
7	$\frac{1}{z^7}+\frac{7}{6 z^5}+\frac{259}{360 z^3}+\frac{5}{16 z}$
8	$\frac{1}{z^8}+\frac{4}{3 z^6}+\frac{14}{15 z^4}+\frac{16}{35 z^2}$
9	$\frac{1}{z^9}+\frac{3}{2 z^7}+\frac{47}{40 z^5}+\frac{3229}{5040 z^3}+\frac{35}{128 z}$
10	$\frac{1}{z^{10}}+\frac{5}{3 z^8}+\frac{13}{9 z^6}+\frac{164}{189 z^4}+\frac{128}{315 z^2}$
11	$\frac{1}{z^{11}}+\frac{11}{6 z^9}+\frac{209}{120 z^7}+\frac{17281}{15120 z^5}+\frac{117469}{201600 z^3}+\frac{63}{256 z}$
12	$\frac{1}{z^{12}}+\frac{2}{z^{10}}+\frac{31}{15 z^8}+\frac{278}{189 z^6}+\frac{3832}{4725 z^4}+\frac{256}{693 z^2}$

2.4 Formulas of Theorem 3

Theorem 3 states that

$$\sum_{k=-\infty}^{\infty} (-1)^{rk} G_{0,r}\left(\frac{1}{z-k\pi}\right) = \frac{1}{\sin^r(z)}.$$

The convergence is absolute in every point $z \in D := \mathbb{C} \setminus \{k\pi \mid k \in \mathbb{Z}\}$ and uniform in every compact subset of D.

The series on the left hand side of the above equation is *the partial fraction decomposition of the function* $\frac{1}{\sin^r}$. It is given by the following *Mathematica* function:

```
PartFrac[r_, z_] := Sum[((-1)^r)^k HoldForm[Evaluate[G[r, 1/(z - k π)]]], {k, -∞, +∞}]
```

$$\text{PartFrac}[r\_, z\_] := \sum_{k=-\infty}^{+\infty} ((-1)^r)^k \,\text{HoldForm}\left[\text{Evaluate}\left[G\left[r, \frac{1}{z-k\pi}\right]\right]\right]$$

For $r = 1$ and $r = 2$ we obtain that :

```
PartFrac[1, z]
```

$$\sum_{k=-\infty}^{\infty} (-1)^k \frac{1}{-k\pi+z}$$

```
PartFrac[2, z]
```

$$\sum_{k=-\infty}^{\infty} \frac{1}{(-k\pi+z)^2}$$

Some other special cases are as follows:

```
TableForm[PartFrac[#, z] & /@ Range[3, 10], TableHeadings → {Range[3, 10], None}]
```

3	$\sum_{k=-\infty}^{\infty} (-1)^k \left(\frac{1}{(-k\pi+z)^3} + \frac{1}{2(-k\pi+z)}\right)$
4	$\sum_{k=-\infty}^{\infty} \left(\frac{1}{(-k\pi+z)^4} + \frac{2}{3(-k\pi+z)^2}\right)$
5	$\sum_{k=-\infty}^{\infty} (-1)^k \left(\frac{1}{(-k\pi+z)^5} + \frac{5}{6(-k\pi+z)^3} + \frac{3}{8(-k\pi+z)}\right)$
6	$\sum_{k=-\infty}^{\infty} \left(\frac{1}{(-k\pi+z)^6} + \frac{1}{(-k\pi+z)^4} + \frac{8}{15(-k\pi+z)^2}\right)$
7	$\sum_{k=-\infty}^{\infty} (-1)^k \left(\frac{1}{(-k\pi+z)^7} + \frac{7}{6(-k\pi+z)^5} + \frac{259}{360(-k\pi+z)^3} + \frac{5}{16(-k\pi+z)}\right)$
8	$\sum_{k=-\infty}^{\infty} \left(\frac{1}{(-k\pi+z)^8} + \frac{4}{3(-k\pi+z)^6} + \frac{14}{15(-k\pi+z)^4} + \frac{16}{35(-k\pi+z)^2}\right)$
9	$\sum_{k=-\infty}^{\infty} (-1)^k \left(\frac{1}{(-k\pi+z)^9} + \frac{3}{2(-k\pi+z)^7} + \frac{47}{40(-k\pi+z)^5} + \frac{3229}{5040(-k\pi+z)^3} + \frac{35}{128(-k\pi+z)}\right)$
10	$\sum_{k=-\infty}^{\infty} \left(\frac{1}{(-k\pi+z)^{10}} + \frac{5}{3(-k\pi+z)^8} + \frac{13}{9(-k\pi+z)^6} + \frac{164}{189(-k\pi+z)^4} + \frac{128}{315(-k\pi+z)^2}\right)$

Further special cases can be calculated.

Mathematica can compute the sums of the above series:

```
Clear[sum]
```

$$\texttt{sum[r\_, z\_] :=} \sum_{k=-\infty}^{+\infty} (-1)^{rk}\, G\left[r, \frac{1}{z - k\pi}\right]$$

```
TableForm[FullSimplify[sum[#, z]] & /@ Range[1, 10],
 TableHeadings → {Range[1, 10], None}]
```

1	Csc[z]
2	Csc[z]^2
3	Csc[z]^3
4	Csc[z]^4
5	Csc[z]^5
6	Csc[z]^6
7	Csc[z]^7
8	Csc[z]^8
9	Csc[z]^9
10	Csc[z]^{10}

3. Some other formulas

3.1 The sums $A(2r, z) := \sum_{k=-\infty}^{\infty} \frac{1}{(z-k\pi)^{2r}}$

In this Section we give different representations for the above sums.

The first one comes from Corollary 1:

$$A[2, z\_] := \frac{1}{\mathrm{Sin}[z]^2}$$

$$A[r\_, z\_] := A[r, z] = \frac{1}{\mathrm{Sin}[z]^r}\left(1 - \mathrm{Sum}\left[\frac{F[r, j]}{j!}\left(1 - \mathrm{Cos}[z]^2\right)^{j/2} \mathrm{Sin}[z]^{r-j} A[r-j, z], \{j, 2, r-2, 2\}\right]\right)$$

For the first few values of r we get :

```
TableForm[Factor[A[#, z]] & /@ Range[2, 12, 2],
 TableHeadings → {Range[2, 12, 2], None}]
```

2	$\mathrm{Csc}[z]^2$
4	$\frac{1}{3}\left(1 + 2\,\mathrm{Cos}[z]^2\right)\mathrm{Csc}[z]^4$
6	$\frac{1}{15}\left(2 + 11\,\mathrm{Cos}[z]^2 + 2\,\mathrm{Cos}[z]^4\right)\mathrm{Csc}[z]^6$
8	$\frac{1}{315}\left(17 + 180\,\mathrm{Cos}[z]^2 + 114\,\mathrm{Cos}[z]^4 + 4\,\mathrm{Cos}[z]^6\right)\mathrm{Csc}[z]^8$
10	$\frac{(62+1072\,\mathrm{Cos}[z]^2+1452\,\mathrm{Cos}[z]^4+247\,\mathrm{Cos}[z]^6+2\,\mathrm{Cos}[z]^8)\,\mathrm{Csc}[z]^{10}}{2835}$
12	$\frac{(1382+35\,396\,\mathrm{Cos}[z]^2+83\,021\,\mathrm{Cos}[z]^4+34\,096\,\mathrm{Cos}[z]^6+2026\,\mathrm{Cos}[z]^8+4\,\mathrm{Cos}[z]^{10})\,\mathrm{Csc}[z]^{12}}{155\,925}$

Using the above *Mathematica* function **A[r,z]** further special cases can be calculated.

The main advantage of the above representation of *A(2r,z)* is that the functions $\sin^{2j}(z)\cdot A(2j, z)$ are algebraic polynomials of the function $\cos^2 z$, consequently their exact lower and upper bounds can be seen very easily.

Mathematica can compute the above sums for specified values of *r:*

$$\mathrm{MathSumA}[r\_, z\_] := \mathrm{Sum}\left[\frac{1}{(z - k\,\pi)^r}, \{k, -\infty, \infty\}\right]$$

The results may be obtained in different forms:

```
TableForm[MathSumA[#, z] & /@ Range[2, 12, 2],
 TableHeadings → {Range[2, 12, 2], None}]
```

2	$\text{Csc}[z]^2$
4	$\frac{1}{3}(2+\text{Cos}[2z])\,\text{Csc}[z]^4$
6	$\frac{1}{60}(33+26\,\text{Cos}[2z]+\text{Cos}[4z])\,\text{Csc}[z]^6$
8	$\frac{(1208+1191\,\text{Cos}[2z]+120\,\text{Cos}[4z]+\text{Cos}[6z])\,\text{Csc}[z]^8}{2520}$
10	$\frac{(78\,095+88\,234\,\text{Cos}[2z]+14\,608\,\text{Cos}[4z]+502\,\text{Cos}[6z]+\text{Cos}[8z])\,\text{Csc}[z]^{10}}{181\,440}$
12	$\frac{(7\,862\,124+9\,738\,114\,\text{Cos}[2z]+2\,203\,488\,\text{Cos}[4z]+152\,637\,\text{Cos}[6z]+2036\,\text{Cos}[8z]+\text{Cos}[10z])\,\text{Csc}[z]^{12}}{19\,958\,400}$

```
TableForm[FullSimplify[MathSumA[#, z]] & /@ Range[2, 12, 2],
 TableHeadings → {Range[2, 12, 2], None}]
```

2	$\text{Csc}[z]^2$
4	$\frac{1}{3}(2+\text{Cos}[2z])\,\text{Csc}[z]^4$
6	$\frac{2\,\text{Csc}[z]^2}{15}-\text{Csc}[z]^4+\text{Csc}[z]^6$
8	$-\frac{4}{315}\text{Csc}[z]^2+\frac{2\,\text{Csc}[z]^4}{5}-\frac{4\,\text{Csc}[z]^6}{3}+\text{Csc}[z]^8$
10	$\frac{2\,\text{Csc}[z]^2}{2835}-\frac{17\,\text{Csc}[z]^4}{189}+\frac{7\,\text{Csc}[z]^6}{9}-\frac{5\,\text{Csc}[z]^8}{3}+\text{Csc}[z]^{10}$
12	$-\frac{4\,\text{Csc}[z]^2}{155\,925}+\frac{62\,\text{Csc}[z]^4}{4725}-\frac{256\,\text{Csc}[z]^6}{945}+\frac{19\,\text{Csc}[z]^8}{15}-2\,\text{Csc}[z]^{10}+\text{Csc}[z]^{12}$

```
TableForm[Factor[FullSimplify[MathSumA[#, z]]] & /@ Range[2, 12, 2],
 TableHeadings → {Range[2, 12, 2], None}]
```

2	$\text{Csc}[z]^2$
4	$\frac{1}{3}(2+\text{Cos}[2z])\,\text{Csc}[z]^4$
6	$\frac{1}{15}\text{Csc}[z]^2\left(2-15\,\text{Csc}[z]^2+15\,\text{Csc}[z]^4\right)$
8	$\frac{1}{315}\text{Csc}[z]^2\left(-4+126\,\text{Csc}[z]^2-420\,\text{Csc}[z]^4+315\,\text{Csc}[z]^6\right)$
10	$\frac{\text{Csc}[z]^2\,(2-255\,\text{Csc}[z]^2+2205\,\text{Csc}[z]^4-4725\,\text{Csc}[z]^6+2835\,\text{Csc}[z]^8)}{2835}$
12	$\frac{\text{Csc}[z]^2\,(-4+2046\,\text{Csc}[z]^2-42\,240\,\text{Csc}[z]^4+197\,505\,\text{Csc}[z]^6-311\,850\,\text{Csc}[z]^8+155\,925\,\text{Csc}[z]^{10})}{155\,925}$

Our formulas for $A(2\,r,\,z)$ are same as the formulas given by *Mathematica:*

```
Table[FullSimplify[MathSumA[r, z] - A[r, z]], {r, 2, 12, 2}]
```

```
{0, 0, 0, 0, 0, 0}
```

3.2 The sums $B\,(2\,r-1,\,z) := \sum_{k=-\infty}^{\infty} \frac{(-1)^k}{(z-k\pi)^{2\,r-1}}$

In this Section we give different representations for the above sums.

The first one comes from Corollary 2:

```
B[1, z_] := 1/Sin[z]
B[r_, z_] := B[r, z] =
  1/Sin[z]^r (1 - Sum[F[r, j]/j! (1 - Cos[z]^2)^(j/2) Sin[z]^(r-j) B[r - j, z], {j, 2, r, 2}])
```

For the first few values of *r* we get :

```
Table[Factor[B[r, z]], {r, 1, 11, 2}] // TableForm
```

$$\text{Csc}[z]$$

$$\frac{1}{2}\left(1+\text{Cos}[z]^2\right)\text{Csc}[z]^3$$

$$\frac{1}{24}\left(5+18\,\text{Cos}[z]^2+\text{Cos}[z]^4\right)\text{Csc}[z]^5$$

$$\frac{1}{720}\left(61+479\,\text{Cos}[z]^2+179\,\text{Cos}[z]^4+\text{Cos}[z]^6\right)\text{Csc}[z]^7$$

$$\frac{\left(1385+19\,028\,\text{Cos}[z]^2+18\,270\,\text{Cos}[z]^4+1636\,\text{Cos}[z]^6+\text{Cos}[z]^8\right)\text{Csc}[z]^9}{40\,320}$$

$$\frac{\left(50\,521+1\,073\,517\,\text{Cos}[z]^2+1\,949\,762\,\text{Cos}[z]^4+540\,242\,\text{Cos}[z]^6+14\,757\,\text{Cos}[z]^8+\text{Cos}[z]^{10}\right)\text{Csc}[z]^{11}}{3\,628\,800}$$

Using the above *Mathematica* function **B[r,z]** further special cases can be calculated.

```
Clear[MathSumB]
MathSumB[r_, z_] := Sum[(-1)^k/(z - k π)^r, {k, -∞, ∞}]
```

The results may be obtained in different forms:

```
TableForm[MathSumB[#, z] & /@ Range[1, 7, 2],
 TableHeadings → {Range[1, 7, 2], None}]
```

1	$\text{Csc}[z]$
3	$\frac{1}{8}\left(\text{Cot}\left[\frac{z}{2}\right]\text{Csc}\left[\frac{z}{2}\right]^2+\text{Sec}\left[\frac{z}{2}\right]^2\text{Tan}\left[\frac{z}{2}\right]\right)$
5	$\frac{1}{96}\left(\text{Cot}\left[\frac{z}{2}\right]^3\text{Csc}\left[\frac{z}{2}\right]^2+2\,\text{Cot}\left[\frac{z}{2}\right]\text{Csc}\left[\frac{z}{2}\right]^4+2\,\text{Sec}\left[\frac{z}{2}\right]^4\text{Tan}\left[\frac{z}{2}\right]+\text{Sec}\left[\frac{z}{2}\right]^2\text{Tan}\left[\frac{z}{2}\right]^3\right)$
7	$\frac{2\,\text{Cot}\left[\frac{z}{2}\right]^5\text{Csc}\left[\frac{z}{2}\right]^2+26\,\text{Cot}\left[\frac{z}{2}\right]^3\text{Csc}\left[\frac{z}{2}\right]^4+17\,\text{Cot}\left[\frac{z}{2}\right]\text{Csc}\left[\frac{z}{2}\right]^6+17\,\text{Sec}\left[\frac{z}{2}\right]^6\text{Tan}\left[\frac{z}{2}\right]+26\,\text{Sec}\left[\frac{z}{2}\right]^4\text{Tan}\left[\frac{z}{2}\right]^3+2\,\text{Sec}\left[\frac{z}{2}\right]^2\text{Tan}\left[\frac{z}{2}\right]^5}{5760}$

```
TableForm[Simplify[MathSumB[#, z]] & /@ Range[1, 11, 2],
 TableHeadings → {Range[1, 11, 2], None}]
```

1	$\text{Csc}[z]$
3	$\frac{1}{4}(3+\text{Cos}[2\,z])\,\text{Csc}[z]^3$
5	$\frac{1}{192}(115+76\,\text{Cos}[2\,z]+\text{Cos}[4\,z])\,\text{Csc}[z]^5$
7	$\frac{(11\,774+10\,543\,\text{Cos}[2\,z]+722\,\text{Cos}[4\,z]+\text{Cos}[6\,z])\,\text{Csc}[z]^7}{23\,040}$
9	$\frac{(2\,337\,507+2\,485\,288\,\text{Cos}[2\,z]+331\,612\,\text{Cos}[4\,z]+6552\,\text{Cos}[6\,z]+\text{Cos}[8\,z])\,\text{Csc}[z]^9}{5\,160\,960}$
11	$\frac{(763\,546\,234+906\,923\,282\,\text{Cos}[2\,z]+178\,300\,904\,\text{Cos}[4\,z]+9\,116\,141\,\text{Cos}[6\,z]+59\,038\,\text{Cos}[8\,z]+\text{Cos}[10\,z])\,\text{Csc}[z]^{11}}{1\,857\,945\,600}$

```
TableForm[FullSimplify[MathSumB[#, z]] & /@ Range[1, 11, 2],
 TableHeadings → {Range[1, 11, 2], None}]
```

1	$\text{Csc}[z]$
3	$-\frac{\text{Csc}[z]}{2}+\text{Csc}[z]^3$
5	$\frac{\text{Csc}[z]}{24}-\frac{5\,\text{Csc}[z]^3}{6}+\text{Csc}[z]^5$
7	$-\frac{\text{Csc}[z]}{720}+\frac{91\,\text{Csc}[z]^3}{360}-\frac{7\,\text{Csc}[z]^5}{6}+\text{Csc}[z]^7$
9	$\frac{\text{Csc}[z]}{40320}-\frac{41\,\text{Csc}[z]^3}{1008}+\frac{23\,\text{Csc}[z]^5}{40}-\frac{3\,\text{Csc}[z]^7}{2}+\text{Csc}[z]^9$
11	$-\frac{\text{Csc}[z]}{3628800}+\frac{7381\,\text{Csc}[z]^3}{1814400}-\frac{2497\,\text{Csc}[z]^5}{15120}+\frac{121\,\text{Csc}[z]^7}{120}-\frac{11\,\text{Csc}[z]^9}{6}+\text{Csc}[z]^{11}$

Our formulas for $B(2\,r-1,\,z)$ are same as the formulas given by *Mathematica:*

```
Table[FullSimplify[MathSumB[r, z] - B[r, z]], {r, 1, 11, 2}]
```

```
{0, 0, 0, 0, 0, 0}
```